
Лабораторная работа №1 

Моделирование объемного n-канального MOSFET 

 

ЗАДАНИЕ: Провести моделирование по индивидуальному заданию планарного 

полевого транзистора. Построить ВАХ прибора. Освоить алгоритм типового моделирования 

в САПР TCAD . 

MOSFET (от слов «металл-оксид-полупроводник-управляемый-полем», англ. metal-oxide-

semiconductor field effect transistor) - полевых транзисторов с изолированным затвором.  

В отличие от биполярных транзисторов, которые управляются током, транзисторы с 

изолированным затвором управляются напряжением, так как, по причине изолированного 

управляющего электрода (затвора) такие транзисторы обладают очень высоким входным 

сопротивлением. 

На рис.1 показана одна из возможных схем поперечного сечения n-канального объемного 

MOSFET с параметрами: длина канала L = 80 нм, толщина диэлектрика tox = 2 нм, глубина 

залегания перехода Xj = 30 нм, исток и сток n+ типа с равномерной концентрацией примеси 10
20

 

см
-3

, подложка р-типа с концентрацией 10
18

 см
-3

. Для моделирования структуры воспользуемся 

САПР TCAD. 

 
Рис. 1. n-канальный объемный MOSFET 

 

Пошаговый алгоритм моделирования: 

1. Создание структуры прибора с использованием ATHENA/ATLAS/DEVEDIT. 

1.1: simulator specification – определение используемого пакета для моделирования 

1.2: mesh definition – задание сетки 

1.3: region definition – задание областей 

1.4: electrode specification - задание электродов 

1.5: doping specification - легирование 

1.6: contact specification – задание контактов 

2. Задание материалов для моделирования. 

3. Задание методики расчета. 

4. Получение предварительного расчета. 

5. Запуск моделирования для получения расчетов при изменении параметров 

моделирования. 

6. Вывод результатов на экран. 

 

Программа для расчета объемного n-канального MOSFET 

 

# Программа для расчета объемного n-канального MOSFET 

# В редакторе DECKBUILD # обозначает комментарий, а не часть программы 

# Шаг 1: Создание структуры прибора 

# 1.1 в каком приложении строим модель 

go atlas 

 

# 1.2 задание сетки 

mesh space.mult=1.0 

# задание сетки по оси х 

# loc обозначает местоположение и определяет положение линии сетки 

x.mesh loc=0.00 spac=0.01 

 

# spac означает шаг и определяет шаг сетки в заданной области 

x.mesh loc=0.05 spac=0.001 

x.mesh loc=0.09 spac=0.004 

x.mesh loc=0.13 spac=0.001 

x.mesh loc=0.18 spac=0.01 

 

# задание сетки по оси у 

y.mesh loc=-0.002 spac=0.0005 

y.mesh loc=0 spac=0.0004 

y.mesh loc=0.03 spac=0.008 

y.mesh loc=0.10 spac=0.01 

 

# 1.3 задание областей прибора 

region num=1 y.min=0 silicon 

region num=2 y.max=0 oxide 

 

# 1.4 задание электродов 

electrode name=gate number=1 x.min=0.05 x.max=0.13 top 

electrode name=source number=2 left length=0.05 y.min=0 y.max=0 

electrode name=drain number=3 right length=0.05 y.min=0 y.max=0 

electrode name=substrate number=4 bottom 

 

# 1.5 определение степени легирования областей, распределение примеси постоянно с 

концентрацией conc=2e18 p.type region=1 

doping uniform conc=2e18 p.type region=1 

doping uniform conc=1e20 n.type x.left=0 x.right=0.05 y.min=0 y.max=0.03 

doping uniform conc=1e20 n.type x.left=0.13 x.right=0.18 y.min=0 y.max=0.03 

# 1.6 определение контактов 

# n.poly задает n+ легированный поликремний как материал контакта с работой выхода 

workfuction=4.17eV 

contact name=gate n.poly 

contact name=source neutral 

contact name=drain neutral 

contact name=substrate neutral 

 

#Шаг 2: Установка материалов модели  

models mos print 

 

#Шаг 3: Определение особенностей модели при расчете (задание метода расчета)  

method newton trap 

 

#Шаг 4: Задание начального приближения при нулевом смещении 

solve init 

 

#Шаг 5: Запуск моделирования для получения графиков при разных смещениях затвора 

#расчет нескольких кривых токов при изменении напряжения сток-исток 

solve vdrain=0.1 outf=solve_vdrain1 

solve vdrain=0.2 outf=solve_vdrain2 

solve vdrain=0.3 outf=solve_vdrain3 

solve vdrain=0.4 outf=solve_vdrain4 

 

# степень изменения напряжения на затворе на отдельных кривых ток стока от напряжения 

сток-исток 

load infile=solve_vdrain1 

 

# вывод результатов в специальный log файл 

log outf=gate1.log 

solve name=gate vgate=0 vfinal=1.2 vstep=0.1 

 

load infile=solve_vdrain2 

log outf=gate2.log 

solve name=gate vgate=0 vfinal=1.2 vstep=0.1 

 

load infile=solve_vdrain3 

log outf=gate3.log 

solve name=gate vgate=0 vfinal=1.2 vstep=0.1 

 

load infile=solve_vdrain4 

log outf=gate4.log 

solve name=gate vgate=0 vfinal=1.2 vstep=0.1 

 

#Шаг 6: Вывод результатов на экран 

# вывод всех log файлов на одном графике 

tonyplot -overlay gate1.log gate2.log gate3.log gate4.log 

quit 

 

На рис. 2 показаны кривые зависимости Id-Vgs MOSFET структуры при напряжениях сток-

исток 0,1, 0,2, 0,3 и 0,4 В. 

 

 
 

Рис 2. Кривые Id-Vgs (ВАХ) MOSFET структуры с L = 80 нм, tox = 2 нм 

Лабораторная работа №2 

Моделирование КНИ МОП-структуры 

 

ЗАДАНИЕ: Провести моделирование по индивидуальному заданию КНИ полевого 

транзистора. Построить ВАХ прибора. Научиться выводить структуры прибора (состав 

прибора, распределение по структуре концентрации примеси). 

Полностью обедненный КНИ транзистор – это прибор, изготовленный на кремнии с 

использованием толстого слоя SiO2 в качестве изолирующей подложки, на которой помещается 

область обеднения. КНИ технология может оказаться хорошей заменой классической планарной 

технологии. Толстый слой диоксида кремния является распространенным углубленным оксидным 

слоем (УОС). Область кремния над УОС – это область, в которой формируется приборная 

структура. Полностью обедненная КНИ структура имеет значительные преимущества над 

объемным кремниевым МОП, поскольку снижается емкость между стоком (истоком) и объемом 

прибора, уменьшаются токи утечки через затвор, убираются наводки; малые потери на 

диэлектрике вместе с большим сопротивлением КНИ подложек нашли широкое распространение 

в радиотехнике и при защите от воздействия радиации. КНИ структуры не только предотвращают 

«защелкивание» и улучшают защиту от ошибок в цифровых схемах, они также позволяют лучше 

контролировать канал, что улучшает подпороговый наклон и снижает эффекты короткого канала. 

Однако КНИ структуры подвержены некоторым динамическим эффектам плавающего 

напряжения. В этом примере с помощью ATLAS моделируется вариант КНИ МОП структуры, в 

которой возможно изменять параметры: длину затвора, концентрацию примесей и т.д. 

Для моделирования используются блоки DECKBUILD, ATLAS и TONYPLOT. 

 

Пошаговый алгоритм для моделирования КНИ МОП структуры. 

1. Задаем и инициализируем используемые переменные. 

Переменные могут быть полезны для изменения конкретных параметров, когда код 

запущен с уже заданными начальными условиями. Для повторного построения модели 

необходимо ввести команды, приведенные в теле программы ниже. 

 

# Программа для моделирования КНИ структуры 

# В DECKBUILD открываем ATLAS 

go atlas 

 

#определение средней точки кремния толщиной sMp=tSi/2 

set sMp=0.0125 

#определение нижней границы кремния толщиной sTsi=tSi 

set sTsi=0.025 

#определение нижней границы слоя оксида заданной толщины 

set sToxb=0.1 

#определение начала верхнего (затворного) слоя оксида заданной толщины 

set sToxf=-0.002 

#минимальное положение электрода затвора по оси х 

set gmin=1 

# максимальное положение электрода затвора по оси х  

set gmax=2 

 

2. Определяем схему структуры. 

Первым шагом при определении структуры КНИ прибора является задание блока схемы с 

большинством интересующих параметров. Первой командой будет команда определения сетки 

прибора. Эта команда скажет ATLASу коэффициент пересчета для создаваемой сетки. В нашем 

случае прибор прямоугольный. По умолчанию симметрия ячейки прямоугольная. 

Ячейка будет более маленькой для большего коэффициента и более большой для меньшего 

коэффициента сетки. Стандартный параметр установлен равным 1. Параметры ячейки вводятся 

как вертикальные и горизонтальные линии в мкм в виде расстояния от центральной линии. ATLAS 

использует сетку с треугольными ячейками. 

 

mesh space.mult=1.0 

#задание сетки по оси х 

x.mesh loc=0.00 spac=0.50 

x.mesh loc=1.15 spac=0.02 

x.mesh loc=1.5 spac=0.1 

x.mesh loc=1.85 spac=0.02 

x.mesh loc=3 spac=0.5 

 

# задание сетки по оси у 

y.mesh loc=$sToxf spac=0.02 

y.mesh loc=0.00 spac=0.005 

y.mesh loc=$sMp spac=0.02 

y.mesh loc=$sTsi spac=0.01 

y.mesh loc=$sToxb spac=0.25 

 

3. Определение различных областей и присвоение им соответствующих номеров. 

После определения ячеистой структуры необходимо задать области прибора. В случае КНИ 

МОП используются три области: (1) область подзатворного оксида для изоляции контакта затвора, 

(2) область кремния толщиной tSi и (3) углубленный оксид (УОС). 

Эти области будут использованы для задания материалов и их параметров в приборе. 

Области должны задаваться по границам ячейки и параметры должны быть аналогичны 

используемым при задании ячейки (выше). ATLAS позволяет пользователю задавать порядка 200 

различных областей в приборе. Если происходит пересечение двух областей, ATLAS будет 

использовать тип материала из области, которая была задана последней. Для того, чтобы ATLAS 

заработал, необходимо полностью описать все области двумерной структуры. 

 

# определение областей 

region number=1 x.min=0 x.max=3 y.min=$sToxf y.max=0 material=Oxide 

region number=2 x.min=0 x.max=3 y.min=0 y.max=$sTsi material=Silicon 

region number=3 x.min=0 x.max=3 y.min=$sTsi y.max=$sToxb material=Oxide 

 

4. Определение имен каждого электрода. 

Для того чтобы показать программе место расположения контактов затвора, истока и стока 

необходимо определить эти электроды в коде. Можно предположить, что возникнет 

несоответствие между технологией изготовления прибора и программированием в ATLASе (т.к. 

верхний окисел и электроды определяются раньше, чем легируется кремний). ATLAS 

предполагает, что структура уже определена. Поскольку электрод затвора находится выше 

верхнего диэлектрика, оба значения y.min и y.max могут быть соотнесены с sTOXF(-0.002), 

который переместит их поверх оксида, но не задаст толщину. Аналогично, контакты к подложке 

располагается под нижним оксидом. 

 

# определение электродов 

electrode name=gate number=1 x.min=$gmin x.max=$gmax y.min=$sToxf y.max=$sToxf 

electrode name=source number=2 x.min=0 x.max=0.5 y.min=0 y.max=0 

electrode name=drain number=3 x.min=2.5 x.max=3 y.min=0 y.max=0 

electrode name=substrate number=4 x.min=0 x.max=3 y.min=$sToxb y.max=$sToxb 

 

5. Задание «контактов» и параметров границ. 

Команда «contact» используется для того, чтобы сказать ATLASу как задействовать 

электроды. По умолчанию контакт к электроду омический. Если разработчик хочет получить 

контакт Шоттки, он должен использовать wirkfunction. Материал контакта затвора 

устанавливается к n
+
 легированному поли кремнию параметром (n.poly). В другом случае (если не 

используется материал (n.poly), workfunction будет определен установкой параметра требуемого 

значения. Например, фраза «contact name=gate workfunction=4.17» эквивалентна фразе «contact 

name=gate n.poly», поскольку работа выхода (work function) из поликремния равна 4.17 эВ. 

Процесс производства, независимо от того, кок его проводить, задействует параметры на 

границе SiO2-Si, что значительно влияет на электрические параметры прибора. Состояния на 

границе, не важно - врожденные, полученные при проведении процессов или возникшие при 

работе, оказываются причиной деградации параметров прибора (крутизна характеристики, 

подвижность носителей, пороговое напряжение) и сильно снижает надежность и время жизни 

прибора. Параметр «interface» используются для определения плотности зарядов на границе 

полупроводник-диэлектрик. Он показывает, что вся граница между полупроводником и 

диэлектриком имеет постоянный заряд 3×10
10

 Кл/см
2
. 

 

contact name=gate n.poly 

interface qf=3e10 

contact name=source neutral 

contact name=drain neutral 

contact name=substrate neutral 

 

6. Определение параметров легирования МОП структуры. 

Следующим действием является введение примеси. Silvaco позволяет разработчику 

задавать тип легирующей примеси и концентрацию. Оно также задать распределение примеси. 

ATLAS может распределять примеси равномерно, по методу Гаусса или другому 

поддерживаемому профилю. Для нашего прибора все примеси будут определены в области 2, в 

которой расположен Si. В начале задается полное легирование p-типом на всей области. За ним 

следует легирование распределением Гаусса носителей n-типа для создания областей истока и 

стока. Концентрации, записанные в команде легирования по Гауссу для истока и стока имеют 

максимум = 1×10
20

 см
-3

 при y = 0 и параметр = 0.05, который является характеристикой длины 

имплантации (стандартное отклонение), для которой уровень легирования вертикально упадет. 

Поперечное падение уровня легирования по оси х определяется параметром lat.char. 

 

#легирование р-типом с равномерным распределением концентрации по всему кремнию 

doping uniform conc=2e17 p.type direction=y regions=2 

#легирование профилем Гаусса области истока 

doping gaussian characteristic =.05 conc=1e20 n.type x.left=0 x.right=$gmin y.top=0 

lat.char=0.05 direction=y 

# легирование профилем Гаусса области стока 

doping gaussian characteristic =.05 conc=1e20 n.type x.left=$gmax x.right=3 y.top=0 lat.char 

=.05 direction=y 

 

7. Сохранение и вывод программы на дисплей. 

Параметр «master» после имени выходного файла означает, что выходной файл будет 

записан как файл стандартной структуры, а не в бинарном формате. Изготовленная МОП 

структура выводится с помощью TonyPlot. 

 

struct outf=SOI.str master 

tonyplot SOI.str 

 

8. Выбор моделей. 

В нашей модели для определения стандартной зависимости подвижности от концентрации 

(conmob) и подвижности от параллельного поля (fldmob) выбрано влияние скорости насыщения. 

Для статистики носителей заряда выбраны функция Ферми-Дирака и сужение запрещенной зоны. 

Параметры evsatmod и hvsatmod с присоединенными b.electrons и b.holes используются для 

детального определения зависимости от приложенного поля. Генерация и рекомбинация – это 

процессы, посредством которых полупроводниковый материал выводится из термодинамического 

равновесия и возвращается назад. Выбраны модели для рекомбинации Шокли-Рида-Холла (SRH) 

и Аугера. 

При наличии сильного легирования (более 10
18

 см
-3

) р-n переход в кремнии становится 

зависимым от количества примесей. При росте легирования уменьшается запрещенная зона, 

поскольку зона проводимости снижается настолько, насколько поднимется валентная зона.  

 

# определение модели 

models auger srh conmob fldmob b.electrons=2 b.holes=1 evsatmod=0 hvsatmod=0 bgn 

temperature=300 

 

9. Определение численных методов. 

Метод Ньютона выбирается с максимальным приближением до 25. Для решения проблемы 

расхождения решений с малым начальным приближением используется параметр «trap», в 

котором maxtrap – это максимально возможное количество «хвостов» (по умолчанию = 4) с 

наклоном шага, уменьшаемым параметром, известным как atrap. Для улучшения характеристик 

при медленной сходимости используется метод Ньютона-Ричардсона (вариация метода Ньютона), 

задаваемый параметром autonr. 

 

# определение метода  

method newton itlimit=25 trap atrap=0.5 maxtrap=4 autonr 

 

10. Определение заданных смещений. 

Для определения профиля легирования при нулевом смещении необходимо определить 

начальные приближения напряжений и концентрации носителей посредством параметра «solve 

init», если нет предварительных вычислений. Если проектировщик пропустит это условие, ATLAS 

автоматически задаст их перед получением первого «solve». 



11 

 

Сложнее всего получить хорошее схождение двух первых ненулевых параметров, 

поскольку используется расчет при нулевом смещении, задаваемый с помощью «solve init» как 

мягкое начальное приближение. Поэтому для расчета двух первых ненулевых параметров 

необходимо использовать небольшое напряжение. 

 

# Установить начальные значения и рассчитать  

# посчитать прибор при нулевом смещении 

solve init 

# посчитать прибор при изменении напряжения сток-исток сначала с шагом 0.01, далее с 

шагом 0.05 и наконец, с шагом 0.1 

 

# два первых ненулевых расчета при малых смещениях сток-исток 

solve vdrain=0.01 

solve vdrain=0.02 

solve vdrain=0.1 

 

#Получение Id-Vds характеристик при разных Vgs 

solve vgate=1.0 outf=solve_vgate1 

solve vgate=1.5 outf=solve_vgate2 

solve vgate=2.0 outf=solve_vgate3 

solve vgate=2.5 outf=solve_vgate4 

 

load infile=solve_vgate1 

log outf=SOI11.log 

solve name=drain vdrain=0 vfinal=2.0 vstep=0.1 

 

load infile=solve_vgate2 

log outf=SOI21.log 

solve name=drain vdrain=0 vfinal=2.0 vstep=0.1 

 

load infile=solve_vgate3 

log outf=SOI31.log 

solve name=drain vdrain=0 vfinal=2.0 vstep=0.1 

 

load infile=solve_vgate4 

log outf=SOI41.log 

solve name=drain vdrain=0 vfinal=2.0 vstep=0.1 

Рис. 3. Схема поперечного сечения полностью обедненного n-канального КНИ МОП 

 

11. Вывод на экран результатов моделирования. 

 

# Отображение на графике всех log файлов  

tonyplot -overlay SOI11.log SOI21.log SOI31.log SOI41.log 

# выход из программы 

quit 

Рис. 4. График зависимости тока стока от напряжения сток-исток при разных напряжениях 

на затворе Vgs = 1.0В, 1.5В, 2.0В и 2.5В, соответственно, для n-канального полностью 

обедненного КНИ MOSFET (оси менять вручную) 

 

 
Рис. 5. Распределение концентраций по структуре 

 

 

 
Рис. 6. Структура КНИ MOSFET 

Лабораторная работа №3 

Моделирование 0,18 мкм объемного n-МОП транзистора с круговой имплантацией 

 

ЗАДАНИЕ: Провести моделирование по индивидуальному заданию 0,18 мкм 

объемный n-МОП транзистор с круговой имплантацией. Построить ВАХ прибора. Освоить 

работу в редакторе технологического моделирования Athena.  

Моделируется прибор LDD 0,18 мкм n-канальный МОП (со слабо легированными истоком 

и стоком). Прибор был изготовлен на кремниевой подложке с ориентацией <100>. Подложка р-

типа была получена легированием бором 3×10
15

 см
-3

. Область n
-
 - типа формируется имплантацией 

дозой 10
15

 см
-3

, область n
+
 - типа - 5×10

15
 см

-3
 мышьяка. Ячейка определяется для половины 

симметричной МОП структуры. После всех шагов изготовления структура будет отзеркалена для 

изготовления полного МОП транзистора. Новые шаги созданы для моделирования n-канального 

МОП сразу после извлечения параметров МОП. Состояние МОП структуры и влияние 

последующих шагов процесса можно отобразить вставкой команды для прорисовки структуры 

(tonyplot structure_name.str) после каждого шага. 

Пошаговый алгоритм моделирования. 

1. Задание ячейки и описание структуры. 

# начало работы в симуляторе технологических процессов ATHENA 

go Athena 

# Задание сетки по оси х 

line x loc=0.15 spac=0.1 

line x loc=0.2 spac=0.006 

line x loc=0.4 spac=0.006 

line x loc=0.6 spac=0.01 

# Задание сетки по оси у 

line y loc=0.0 spac=0.002 

line y loc=0.2 spac=0.005 

line y loc=0.5 spac=0.05 

line y loc=0.8 spac=0.15 

# structure declaration 

struct outfile=nmos_bulk.str 

 

2. Задание кремниевой подложки с ориентацией 100 и легированием бором 10
15

 см
-3

. 

Пространственный умножитель (Space.mult = 2) используется для ускорения процесса. 

Уменьшение параметра пространственного умножителя используется для получения большей 

точности. 

 

init orientation=100 c.boron=1.0e15 space.mul=2 

 

3. Выращивание тонкого гладкого слоя SiO2 толщиной от 10 до 15 нм на подложке для 

уменьшения влияния канала и предотвращения загрязнения подложки. Простейший метод 

нанесения в ATHENA – это согласованное (обладающий свойствами постоянства растяжений и 

сохранения углов) нанесение. Оно используется, когда точность формы наносимого слоя не 

критична. Сухое окисление проводится в течение 30 минут при 1000
о
С. Последующее 

стравливание формирует равномерное покрытие оксидом толщиной 0.02 мкм. 

 

diffus time=30 temp=1000 dryo2 press=1.00 hcl=3 

etch oxide thick=0.02 

 

4. Для имплантации бора в размере 8.0×10
12

 ионов/см
2
 с энергией 100 КэВ по умолчанию 

используется двойная модель Пирсона. Канал n-типа в МОП транзисторе должен быть 

сформирован на кремнии р-типа, так как этот материал под затвором должен быть 

противоположного типа проводимости. Следовательно, следующими шагом является имплантация 

бора для формирования р-кармана на подложке. Изготовление n-канального МОП транзистора 

может быть проведено с подложки р-типа, но в классической технологии имплантацией 

формируется карман р-типа. ATHENA предлагает 3 модели имплантации; (1) двойного Пирсона 

(по умолчанию), (2) одинарного Пирсона и (3) Монте-Карло. 

 

implant boron dose=3.0e13 energy=200 pearson 

 

5. Перемещение атомов по решетке и занятие необходимых позиций. В результате ионной 

имплантации максимальное количество примеси оказывается на средней глубине проникновения в 

зависимости от особенностей внедрения примеси. Для того, чтобы увеличить равномерность 

легирования, подложку подогревают до высоких температур, в следствие чего атомы бора 

получают достаточное количество энергии для перемещения по кристаллической решетке и более 

равномерного распределения по подложке. Если подогрев происходит в атмосфере кислорода, то 

формируется оксидная пленка. В данном случае используется влажное окисление. 

 

diffus temp=950 time=100 weto2 hcl=3 

 

6. Дальнейшее перемещение р-области в подложку и увеличение равномерности 

легирования использованием нескольких операций диффузии при разных температурах, что 

позволяет хорошо управлять процессом. 

 

diffus time=50 temp=1000 t.rate=4.000 dryo2 press=0.10 hcl=3 

diffus time=220 temp=1200 nitro press=1 

diffus time=90 temp=1200 t.rate=-4.444 nitro press=1 

 

7. Удаление всех слоев оксидов для получения поверхности, на которой будет проходить 

процесс задания физических параметров MOSFET. 

 

etch oxide all 

 

8. Стравливание поверхности для того, чтобы поверхность не загрязнялась из-за 

предыдущих операций. В этом процессе тонкий слой поликремния сначала окисляется, а затем 

полученный оксид удаляется. 

 

diffus time=20 temp=1000 dryo2 press=1 hcl=3 

etch oxide all 

 

9.Фоормирование подзатворного диэлектрика окислением в сухом кислороде. Толщина 

оксида является важным параметром при определении характеристик MOSFET и контролируется 

временем и температурой осаждения. 

diffus time=3 temp=895 dryo2 press=1.00 hcl=1 

 

10. Задание порогового напряжения имплантацией бора через подзатворный диэлектрик. 

Увеличение дозы имплантации бора приводит к увеличению порогового напряжения, 

инвертировать канал р-типа сложнее. 

 

implant boron dose=1.5e13 energy=45 pearson 

 

11. Конформное (обладающий свойствами постоянства растяжений и сохранения углов) 

нанесение поликремния для создания затвора. 

 

depo poly thick=0.2 divi=10 

 

12. Начало формирования рельефа поликремниевого затвора стравливанием 0,35 мкм слева. 

 

etch poly left p1.x=0.51 

 

13. Имплантация через нанесенный диэлектрик t областей истока/стока. 

 

method fermi compress 

diffuse time=5 temp=900 weto press=0.8 

implant arsenic dose=1.0e15 energy=30 pearson 

 

14. Имплантация кольца р-типа. 

 

implant boron dose=3.0e13 energy=15 tilt=30 fullrotat 

 

15. Изготовление оксидного ограничителя для того, чтобы обеспечить изоляцию и 

сохранить границы для дальнейшей имплантации. 

 

depo oxide thick=0.10 divisions=8 

etch oxide dry thick=0.10 

 

16. Формирование сильно-легированных областей истока и стока имплантацией мышьяка 

вместо фосфора, который используется для изготовления слабо-легированных истока и стока. 

 

implant arsenic dose=5e15 energy=60 pearson 

 

17. Отжиг в атмосфере азота для диффузии сформированных ранее истока и стока. 

 

method fermi compress 

diffuse time=1 temp=1000 nitro press=1.0 

 

18. Удаление оксида над областями истока и стока для нанесения металлических контактов 

к истоку и стоку. 

etch oxide left p1.x=0.35 

 

19. Нанесение алюминия в качестве низкоомных контактов. 

 

deposit alumin thick=0.03 divi=2 

 

20. Удаление излишков металла 

 

etch alumin right p1.x=0.33 

 

21. «Отзеркаливание» структуры для получения симметричного прибора 

 

structure mirror right 

 

22. Определение местоположения электродов и сохранение собранной структуры 

 

electrode name=gate x=0.59 y=0.1 

electrode name=source x=0.2 

electrode name=drain x=1.0 

electrode name=substrate backside 

save outfile=nmos_bulk.str 

 

23. «Извлечение» параметров МОП структуры и построение графика 

 

# расчет толщины подзатворного диэлектрика 

extract name=“gateox” thickness oxide mat.occno=1 x.val=0.59 

 

# расчет глубины залегания p-n перехода Xj  

extract name=“nxj” xj silicon mat.occno=1 x.val=0.2 junc.occno=1 

 

# расчет поверхностного сопротивления n++ областей 

extract name=“n++ sheet rho” sheet.res material=“Silicon” mat.occno=1 x.val=0.2 

region.occno=1 

 

# расчет слоя rho под защитным покрытием, области LDD  

extract name=“ldd sheet rho” sheet.res material=“Silicon” mat.occno=1 x.val=0.49 

region.occno=1 

 

# расчет поверхностной концентрации в канале 

extract name=“chan surf conc” surf.conc impurity=“Net Doping” material=“Silicon” mat.occno=1 

x.val=0.6 

 

# построение графиков структуры 

tonyplot nmos_bulk.str 

 

Программа для расчета различных параметров, порогового напряжения и Ids-Vgs 

характеристик работает со структурой, сгенерированной в блоке ATHENA. Пороговое напряжение 

рассчитывается с помощью определения пересечения максимального наклона кривой Ids-Vgs с 

осью х. Расчётные значения выводятся далее в нижней части окна DECKBUILD после запуска 

кода на генерацию модели. 

 

go atlas 

# импортирование структуры для автоматического пересчета из ATHENA в ATLAS 

mesh inf=nmos_bulk.str 

 

# задание модели 

models cvt srh numcarr=2 

 

# особенности методики расчета 

method newton itlimit=25 trap atrap=0.5 maxtrap=4 autonr nrcriterion=0.1 tol.time=0.005 

dt.min=1e-25 

 

# расчет зависимости Id-Vgs с изменением напряжения на затворе и сохранение результатов 

в log файл  

solve init 

solve vdrain=0.1 

log outf=bulkHalo.log master 

solve name=gate vgate=0.1 vfinal=1.5 vstep=0.1 

 

# построение графика log файла 

tonyplot bulkHalo.log 

 

# расчет порогового напряжения 

extract name=“vt” (xintercept(maxslope(curve(abs(v.”gate”),abs(i.”drain”)))) - 

abs(ave(v.”drain”))/2.0) 

quit 

 

Результаты моделирования. 

Рис. 7. Зависимости Id-Vgs (ВАХ) для n-канального МОП транзистора 

 

 
 

Рис. 8. Структура n-канального МОП транзистора 

Рис. 9. Распределение примеси по структуре n-канального МОП транзистора 

Лабораторная работа №4 

Объемная инверсия в двухзатворном МОП-транзисторе (DG MOSFET) 

 

ЗАДАНИЕ: Провести моделирование по индивидуальному заданию двухзатворного 

МОП транзистор. Построить ВАХ прибора. Расчитать характеристики DIBL и SS. 

При достижении КМОП технологией своих пределов масштабирования одним из методов 

решения проблемы изготовления высококачественных маломощных приборов оказывается 

переход к двухзатворным МОП транзисторам (DG MOSFET). Считается, что DG MOSFET (в 

котором канал управляется напряжением на затворе, а не током, текущим от стока к истоку) 

превосходит классические однозатворные МОП структуры по электростатическим параметрам, 

поскольку в нем проводимость канала в значительной степени завязана на контакт затвора. 

 

 
Рис. 10. Размеры структуры двухзатворного МОП транзистора 

 

Использование симметричного DG MOSFET позволяет снизить такие эффекты короткого 

канала, как снижение высоты потенциального барьера при приложении потенциала к стоку (DIBL) 

и ухудшение подпорогового наклона характеристики. В результате исчезает необходимость 

использования сильного легирования канала и значительных градиентов концентраций носителей 

заряда по сравнению с классическими объемными MOSFET. Симметричность DG MOSFET 

говорит о том, что одинаковые напряжения, приложенные к обоим затворам, будут иметь одну и 

ту же работу выхода. Использование собственного полупроводника в качестве объемной области 

транзистора обеспечивает рост подвижности посредством снижения рассеяния заряда на примесях 

и лучшую стабильность параметров транзисторов исключением статистической функции 

концентрации легирующей примеси. Таким образом, в DG MOSFET не используется процесс 

легирования основы структуры, хотя обычно при изготовлении транзистора возникает 

непреднамеренное легирование порядка 10
15

 см
-3

. Таким образом, целью моделирования будет 

создание слаболегированного DG MOSFET. 

В данной работе редактор ATLAS используется для моделирования 50 нм DG n-MOSFET с 

резкими переходами на границах областей истока и стока (Leff = Lmet = 50 нм).  

В качестве параметров структуры заданы: 

- ширина канала W = 50 нм, 

- длина канала L = 50 нм,  

- толщина кремниевого канала tSi = 20 нм, 

- эквивалентная толщина подзатворного диэлектрика tox = 2 нм, 

- концентрация легирующей примеси кремниевого канала NA = 10
15

 см
-3

, 

- концентрация легирующей примеси областей истока и стока ND = 10
20

 см
-3

, 

- металлический затвор со средней работой выхода 4,74 эВ. 

 

Пошаговый алгоритм моделирования. 

# Код создания структуры DG MOSFET 

# начало работы в симуляторе технологических процессов Atlas 

go atlas  

 

mesh space.mult=1.0 

 

# mesh definition in x direction 

x.mesh loc=0.00 spac=0.01 

x.mesh loc=0.05 spac=0.001 

x.mesh loc=0.075 spac=0.005 

x.mesh loc=0.1 spac=0.001 

x.mesh loc=0.15 spac=0.01 

 

# mesh definition in y direction 

y.mesh loc=-0.002 spac=0.001 

y.mesh loc=0 spac=0.0004 

y.mesh loc=0.01 spac=0.001 

y.mesh loc=0.02 spac=0.0004 

y.mesh loc=0.022 spac=0.001 

 

# region definition 

region num=1 y.min=0 y.max=0.02 silicon  

region num=2 y.min=-0.002 y.max=0 oxide  

region num=3 y.min=0.02 y.max=0.022 oxide 

 

# electrode definition 

electrode name=gate number=1 x.min=0.05 x.max=0.1 top  

electrode name=gate1 number=2 x.min=0.05 x.max=0.1 bottom  

electrode name=source number=3 left length=0.05 y.min=0 y.max=0 

electrode name=drain number=4 right length=0.05 y.min=0 y.max=0 

 

# doping specification 

doping uniform conc=1e15 p.type region=1 

doping uniform conc=1e20 n.type x.left=0 x.right=0.05 region=1  

doping uniform conc=1e20 n.type x.left=0.1 x.right=0.15 region=1 

 

# сохранение структуры в файл и вывод данных на экран  

save outfile=dgmos.str  

tonyplot dgmos.str  

quit 

 
 

Рис. 11. Размеры структуры двухзатворного МОП транзистора в TCAD 

 

 

В данной модели не учитываются квантовые эффекты удержания, так как в структуре 

толщина и длина кремниевой пленки больше 10 нм. Квантовое механическое туннелирование от 

истока к стоку (через барьер) ухудшает характеристики прибора в случае короткоканальных 

структур, а в случае канала более 10 нм длиной МОП-транзистор ведет себя классическим 

образом. 

Потенциал поверхности ψS является одним из наиболее подходящих параметров для 

описания физики работы МОП структуры. Он определяется как разница между 

электростатическим потенциалом на границе SiO2/Si и потенциалом нейтральной объемной 

области, возникающих из-за искривления границ энергетических зон.  

 

# Program for obtaining variation of surface and body center potential  

 

go atlas 

 

# IMPORT THE MESH structure  

mesh inf=dgmos.str 

 

# contact specification 

contact name=gate n.poly workfunction=4.74 

 

# two separate electrodes gate and gate1 are shorted by “common” parameter 

contact name=gate1 n.poly workfunction=4.74 common=gate 

contact name=source neutral  

contact name=drain neutral 

 

# model declaration  

models auger srh conmob fldmob bgn temperature=300 

 

# method definition  

method newton itlimit=25 trap 

 

# solution for specific gate and drain bias 

solve init  

solve vdrain=0.0  

solve vgate=0.0 

 

# saving the structure file  

save outf=dg.str 

 

# extraction of surface potential where y.val=0.0 (semiconductor front gate oxide interface) along 

the depth and saving the curve in a file 

extract name=“srp_profile1” curve(depth, potential material=“Silicon” mat.occno=1 y.val=0.0) 

outfile=“extract1.dat” 

 

# extraction of channel center potential where y.val=0.01(center of the SOI silicon semiconductor 

body y=tSi/2) along the depth and saving the curve in a file  

extract name=“srp_profile2” curve(depth, potential material=“Silicon” mat.occno=1 y.val=0.01) 

outfile=“extract2.dat” 

 

# plot the files in the same window overlaid  

tonyplot -overlay extract1.dat extract2.dat  

quit 

 

На рис. 12 показано изменение потенциалов на поверхности и в глубине канала. На рис. 13 

представлена зависимость поверхностного (ψS) потенциала и потенциала глубины канала (ψ0) от 

подаваемого на затвор напряжения Vgs при заданном напряжении сток-исток Vds (в примере Vds = 

0 В). Однако в TCAD есть возможность рассчитывать потенциалы при разных напряжениях на 

затворе и стоке. 

Рис. 12 Распределение потенциалов на поверхности и в глубине канала 

 

 
Рис. 13. Зависимость поверхностного (ψS) потенциала и потенциала глубины канала (ψ0) от 

подаваемого на затвор напряжения Vgs при заданном напряжении сток-исток Vds 

 

На рисунке 13 приводится зависимость поверхностного (ψS) потенциала и потенциала 

глубины канала (ψ0), рассчитанная при изменении напряжения на затворе от 0 до 1,5 В с шагом 0,1 

В с пересчетом на каждом шаге. Видно, что в области перед порогом, в которой заряд 

полупроводника мал, есть объемная инверсия, а потенциалы примерно равны по всей толщине 

кремниевой пленки и оба стремятся к напряжению на затворе. После того, как напряжение на 

затворе становится больше порогового, плотность электронов становится значительной. В 

результате ψS продолжает медленно расти, а ψ0 выходит на насыщение. Подвижные заряды возле 

границ канала экранируют поле затвора от объема канала, так что ψS и ψ0 разделяются 

(отсутствует объемная инверсия). Зависимости потенциалов канала от напряжения на затворе для 

транзистора с одинаковой длиной канала и разными толщинами имеют точку пересечения, 

показанную на рис. 13. 

 

# Program to obtain Id-Vgs characteristics 

 

go atlas 

 

# IMPORT THE MESH  

mesh inf=dgmos.str 

 

# contact specification 

contact name=gate n.poly workfunction=4.74  

contact name=gate1 n.poly workfunction=4.74 common=gate 

contact name=source neutral  

contact name=drain neutral 

 

# model declaration 

models auger srh conmob fldmob bgn temperature=300 

 

# method definition  

method newton itlimit=25 trap 

 

# initial solution and drain bias specification 

solve init  

solve vdrain=0.05  

#solve vdrain=0.1 

#solve vdrain=0.5 

#solve vdrain=1.5 

 

# output log file declaration  

log outf=dg1.log 

 

# solution to obtain solution for id-vgs characteristics by ramping Vgs 

solve name=gate vgate=0 vfinal=1.2 vstep=0.05 

 

# plotting the structure  

tonyplot dg1.log 

quit 

Рис. 14. Результаты моделирования зависимости тока стока от напряжения затвор-исток 

(ID(VGS)), рассчитанные при низком (VDS = 0,05 В) и высоком (VDS = 1,5 В) напряжении сток-исток 

 
Рис. 15. Пример расчета характеристик DIBL и SS 

 

Из полученных кривых, используя пример на рис. 15, определить снижение высоты 

потенциального барьера при приложении потенциала к стоку (DIBL) и ухудшение подпорогового 

наклона характеристики. Смещение по оси напряжений - рассчитывается как отношение разности 

смещения кривых (мВ) к разности подаваемых на сток потенциалов, а величина подпорогового 

наклона кривых (SS), которая определяется как наклон зависимости ID(VGS) в подпороговой 

области. Рассчитывается как увеличение (мВ) напряжения на затворе, которое требуется для 

увеличения тока стока в 10 раз.  

